Calculation of parameters for arbitrary retarders using a sequence of spatial light modulators

Martin J. Booth

This workbook refers to the content of the manuscript “Arbitrary complex retarders using a sequence of spatial light modulators” by Qi Hu, Chao He, and Martin J Booth, University of Oxford.

A selection of randomly chosen retarder properties is used to test the correspondence between the different derivations shown in the manuscript in order to validate the presented results.

Section and equation numbers used here refer to the manuscript.

Define retarder properties

The expressions used in the manuscript are valid for retardance in range $-\frac{\pi}{2} < \Psi \leq \frac{\pi}{2}$, angle in range $-\pi < \theta \leq \pi$, and elliptical parameter $-\frac{\pi}{2} < \alpha \leq \frac{\pi}{2}$

Define elliptical retarder properties: retardance Ψ, orientation θ, elliptical parameter α. Create table of Nsample sets of random sampled data over range of validity, listed in order Ψ, θ, α

```
In[1]:= Nsamples = 10;

In[2]:= Ψθαsamples = Table[{RandomReal[{-π/2, π/2}],
            RandomReal[{-π, π}], RandomReal[{-π/2, π/2}]}, {dummy, 1, Nsamples}];
```

Reference lists of randomly chosen parameters Ψ, θ, α

```
In[3]:= Ψlist  = Ψθαsamples[[All, 1]]

Out[3]= {-0.20064, 1.45344, -1.48768, -1.42422, -0.260246,
         0.313109, -0.400972, -0.0702045, 0.361852, 0.622332}

In[4]:= θlist  = Ψθαsamples[[All, 2]]

Out[4]= {-1.60779, -0.560071, -2.13897, -2.9638,
         0.392827, 0.0198596, 2.64628, 2.66304, 2.57679, -2.56078}

In[5]:= αlist  = Ψθαsamples[[All, 3]]

Out[5]= {1.48547, 0.719531, 0.757004, -0.814232,
         1.00683, -0.489, -1.41777, 1.28259, 0.805716, -1.54159}
```

Calculate SLM retardance settings and corresponding dynamic phase

Use analytical expressions from sections 3 and 4 to derive angle β and then SLM settings Ψ_1, Ψ_2, Ψ_3.

Eq. 25:

In[6]:=
```
β[Ψ_, θ_] := ArcTan[Tan[Ψ / 2] Cos[2 θ]]
```

Eq. 26:

In[7]:=
```
ψ1[α_, β_] := β - α
```

Eq. 19:

In[8]:=
```
ψ2[Ψ_, θ_] := 2 ArcSin[Sin[Ψ / 2] Sin[2 θ]]
```

Eq. 27:

In[9]:=
```
ψ3[α_, β_] := β + α
```

Calculate SLM settings for each set of randomly chosen retarder parameters

In[10]:=
```
ψ1test = ψ1[#[[3]],  β[#[[1]], #[[2]]]] & /@ Ψθαsamples
```

Out[10]=
```
{-1.38542, -0.350091, -0.387489, 0.133903,
 -1.09907, 0.645433, 1.30683, -1.30281, -0.727762, 1.66885}
```

In[11]:=
```
ψ3test = ψ3[#[[3]],  β[#[[1]], #[[2]]]] & /@ Ψθαsamples
```

Out[11]=
```
{1.58552, 1.08897, 1.12652, -1.49456, 0.914579,
 -0.332567, -1.5287, 1.26237, 0.883671, -1.41433}
```

In[12]:=
```
ψ2test = ψ2[#[[1]], #[[2]]] & /@ Ψθαsamples
```

Out[12]=
```
{-0.0148059, -1.28222, -1.32278, -0.458982, -0.183809,
 0.0123825, 0.334679, 0.0573899, -0.326873, 0.569454}
```

Use analytical expressions to find dynamic phase Θ.

Eq. 28:

In[13]:=
```
Θ[β_, θ_] := β + ArcTan[Tan[2 θ] Sin[β]]
```

In[14]:=
```
Θtest = Θ[β[#[[1]], #[[2]]], #[[2]]] & /@ Ψθαsamples
```

Out[14]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Calculate retardance and dynamic phase from Jones matrix expressions

From the expressions in section 2.2, form the Jones matrix for combination of SLMs using ψ_1, ψ_2, ψ_3, calculate eigenvalues, then equivalent retardance.

Eq. 1, 2:

In[15]:=

$$J1[\phi1\_] := \begin{pmatrix} \mathrm{Exp}[i\,\phi1] & 0 \\ 0 & 1 \end{pmatrix};$$

$$J2[\phi2\_] := \begin{pmatrix} \frac{e^{i\,\phi2}}{2} + \frac{1}{2} & \frac{e^{i\,\phi2}}{2} - \frac{1}{2} \\ \frac{e^{i\,\phi2}}{2} - \frac{1}{2} & \frac{e^{i\,\phi2}}{2} + \frac{1}{2} \end{pmatrix};$$

$$J3[\phi3\_] := \begin{pmatrix} \mathrm{Exp}[i\,\phi3] & 0 \\ 0 & 1 \end{pmatrix};$$

```
JT[ϕ1_, ϕ2_, ϕ3_] := J3[ϕ3].J2[ϕ2].J1[ϕ1]
```

Create set of total Jones matrices based upon the previously calculated set of SLM settings.

In[18]:=
```
JTtest = Table[JT[ψ1test[[n]], ψ2test[[n]], ψ3test[[n]]], {n, 1, Nsamples}];
```

Calculate retardance as the argument of the conjugate product of the two eigenvalues
$\Psi = \arg(\mu_1\, \mu_2{}^*)$

In[19]:=
```
ΨJones = ((Arg[#⟦1⟧ Conjugate[#⟦2⟧]]) &[Eigenvalues[#]]) & /@ JTtest
```

Out[19]=
```
{0.20064, 1.45344, 1.48768, -1.42422, -0.260246,
 0.313109, -0.400972, -0.0702045, -0.361852, 0.622332}
```

Compare with input list (note that sign changes can occur through the above calculation because of the random ordering of eigenmodes when the two eigenvalues have the same magnitude).

In[20]:=
```
Ψlist
```

Out[20]=
```
{-0.20064, 1.45344, -1.48768, -1.42422, -0.260246,
 0.313109, -0.400972, -0.0702045, 0.361852, 0.622332}
```

Check dynamic phase Θ.

Eq. 5:

In[21]:=
```
ΘJones = (1/2 Arg[Det[#]]) & /@ JTtest
```

Out[21]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Compare with list from earlier section

In[22]:=
```
Θtest
```

Out[22]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Calculate retardance and dynamic phase from explicit expression

These calculations of retardance use the expressions in Eq. 12 and Eq. 21 of the manuscript, which are direct functions of the three SLM settings $\psi_1,\ \psi_2,\ \psi_3$.

Eq. 12:

In[23]:=
```
Ψslms[ψ1_, ψ2_, ψ3_] := 2 ArcCos[Cos[ψ2 / 2] Cos[(ψ1 + ψ3) / 2]]
```

Eq. 5:

In[24]:=
```
Θslms[ψ1_, ψ2_, ψ3_] := (ψ1 + ψ2 + ψ3) / 2
```

In[25]:=
```
Ψslmstest = Table[Ψslms[ψ1test⟦n⟧, ψ2test⟦n⟧, ψ3test⟦n⟧], {n, 1, Nsamples}]
```

Out[25]=
```
{0.20064, 1.45344, 1.48768, 1.42422, 0.260246,
 0.313109, 0.400972, 0.0702045, 0.361852, 0.622332}
```

Compare with the input list (note that signs above are all positive; this is due to the ambiguity in the calculation of the term acos(±Y)).

In[26]:=
```
Ψlist
```

Out[26]=
```
{-0.20064, 1.45344, -1.48768, -1.42422, -0.260246,
 0.313109, -0.400972, -0.0702045, 0.361852, 0.622332}
```

Check dynamic phase Θ.

In[27]:=
```
Θslmstest = Table[Θslms[ψ1test[[n]], ψ2test[[n]], ψ3test[[n]]], {n, 1, Nsamples}]
```
Out[27]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Compare with list from earlier section

In[28]:=
```
Θtest
```
Out[28]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Calculate retardance and dynamic phase using Pancharatnam calculations

Using spherical triangle expressions for elliptical retarders from Appendix C, input values from α, β, θ and check retardance.

Vector definitions from the manuscript:

Eq. 49:

In[29]:=
```
A = {Cos[2 θ], Sin[2 θ] Cos[α], Sin[2 θ] Sin[α]};
B = {Cos[2 θ], Sin[2 θ] Cos[β], Sin[2 θ] Sin[β]};
CC = {Cos[2 θ], Sin[2 θ] Cos[β], -Sin[2 θ] Sin[β]};
Qd = {-1, 0, 0};
Rd = {0, -1, 0};
```

Here we define f1e = $\phi_1+\phi_2+\phi_3+\phi_4$ and f2e = $\phi_1'+\phi_2'+\phi_3'+\phi_4' = \phi_1'+\phi_2'+\phi_3'-\phi_4$

In[34]:=
```
f1e[θ_, α_, β_] =
 ArcTan[1 + A.B + B.Qd + Qd.A, A.Qd × B] + ArcTan[1 + B.CC + CC.Rd + Rd.B, B.Rd × CC] +
  ArcTan[1 + CC.A + A.Qd + Qd.CC, CC.Qd × A] + ArcTan[1 + CC.B + B.A + A.CC, A.B × CC]
```
Out[34]=
```
ArcTan[1 - 2 Cos[2 θ] + Cos[2 θ]^2 + Cos[α] Cos[β] Sin[2 θ]^2 - Sin[α] Sin[β] Sin[2 θ]^2,
  Cos[β] Sin[α] Sin[2 θ]^2 + Cos[α] Sin[β] Sin[2 θ]^2] +
 ArcTan[1 - 2 Cos[2 θ] + Cos[2 θ]^2 + Cos[α] Cos[β] Sin[2 θ]^2 + Sin[α] Sin[β] Sin[2 θ]^2,
  -Cos[β] Sin[α] Sin[2 θ]^2 + Cos[α] Sin[β] Sin[2 θ]^2] +
 ArcTan[1 + Cos[2 θ]^2 - 2 Cos[β] Sin[2 θ] + Cos[β]^2 Sin[2 θ]^2 - Sin[β]^2 Sin[2 θ]^2,
  2 Cos[2 θ] Sin[β] Sin[2 θ]] + ArcTan[
  1 + 3 Cos[2 θ]^2 + 2 Cos[α] Cos[β] Sin[2 θ]^2 + Cos[β]^2 Sin[2 θ]^2 - Sin[β]^2 Sin[2 θ]^2,
  2 Cos[α] Cos[2 θ] Sin[β] Sin[2 θ]^2 - 2 Cos[β] Cos[2 θ] Sin[β] Sin[2 θ]^2]
```

In[35]:=
```
f2e[θ_, α_, β_] =
  ArcTan[1 + A.B - B.Qd - Qd.A, A.Qd × B] + ArcTan[1 + B.CC - CC.Rd - Rd.B, B.Rd × CC] +
   ArcTan[1 + CC.A - A.Qd - Qd.CC, CC.Qd × A] - ArcTan[1 + CC.B + B.A + A.CC, A.B × CC]
```

Out[35]=
```
ArcTan[1 + 2 Cos[2 θ] + Cos[2 θ]^2 + Cos[α] Cos[β] Sin[2 θ]^2 - Sin[α] Sin[β] Sin[2 θ]^2,
   Cos[β] Sin[α] Sin[2 θ]^2 + Cos[α] Sin[β] Sin[2 θ]^2] +
  ArcTan[1 + 2 Cos[2 θ] + Cos[2 θ]^2 + Cos[α] Cos[β] Sin[2 θ]^2 + Sin[α] Sin[β] Sin[2 θ]^2,
   - Cos[β] Sin[α] Sin[2 θ]^2 + Cos[α] Sin[β] Sin[2 θ]^2] +
  ArcTan[1 + Cos[2 θ]^2 + 2 Cos[β] Sin[2 θ] + Cos[β]^2 Sin[2 θ]^2 - Sin[β]^2 Sin[2 θ]^2,
   2 Cos[2 θ] Sin[β] Sin[2 θ]] - ArcTan[
   1 + 3 Cos[2 θ]^2 + 2 Cos[α] Cos[β] Sin[2 θ]^2 + Cos[β]^2 Sin[2 θ]^2 - Sin[β]^2 Sin[2 θ]^2,
   2 Cos[α] Cos[2 θ] Sin[β] Sin[2 θ]^2 - 2 Cos[β] Cos[2 θ] Sin[β] Sin[2 θ]^2]
```

For calculation of retardance use f1e-f2e

Eq. 51:

In[36]:=
```
fTe[θ_, α_, β_] := f1e[θ, α, β] - f2e[θ, α, β]
```

In[37]:=
```
Ψtrianglestest = fTe[#[[2]], #[[3]],  β[#[[1]], #[[2]]]] & /@ Ψθαsamples
```

Out[37]=
```
{-0.20064, 1.45344, -1.48768, -1.42422, -0.260246,
 0.313109, -0.400972, -0.0702045, 0.361852, 0.622332}
```

Compare with the input list

In[38]:=
```
Ψlist
```

Out[38]=
```
{-0.20064, 1.45344, -1.48768, -1.42422, -0.260246,
 0.313109, -0.400972, -0.0702045, 0.361852, 0.622332}
```

Check dynamic phase Θ.

For calculation of dynamic phase use 0.5(f2e+f1e),

Eq. 52:

In[39]:=
```
dynTe[θ_, α_, β_] := 1/2 (f2e[θ, α, β] + f1e[θ, α, β])
```

In[40]:=
```
Θtrianglestest = dynTe[#[[2]], #[[3]],  β[#[[1]], #[[2]]]] & /@ Ψθαsamples
```

Out[40]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```

Compare against reference list

In[41]:=
```
Θtest
```

Out[41]=
```
{0.0926442, -0.271671, -0.291874, -0.90982, -0.184152,
 0.162624, 0.0564018, 0.00847416, -0.085482, 0.411987}
```